國防科技學術合作協調小組研究計畫成果報告

低掠角入射海波之電磁散射分析

計畫編號： NSC90-2623-7-009-002

執行期間：90 年 1 月 1 日至 90 年 12 月 31 日

計畫主持人：黃瑞彬 博士

執行單位：交通大學電子與資訊研究中心

中華民國 90 年 12 月 31 日

Scattering of plane waves by sea-wave modeled by one dimensionally periodic structure

PI: Ruey Bing Hwang

Associate Professor

Microelectronics and Information Systems Research Center

Chiao-Tung University

Hsinchu, Taiwan

1. Introduction

The scattering characteristics of plane waves by rough surface have attracted continuous attentions for many years. Various numerical and analytical methods were studied to resolve this problem. For example, since the dimension of sea-wave is larger compared with electromagnetic wavelength, many high frequency asymptotic methods based on ray-tracing method could be applied, such as: GO, GTD, UTD. However, at a low grazing angle incidence, these methods have some problems that have been reported in literatures.

On the other hand, recently some numerical methods were utilized to resolve the scattering characteristics of sea-wave, in particular for the case of low grazing angle incidence. For instance, many researchers use method of moment to investigate the validity of approximate electromagnetic rough-surface scattering theories. However, the commonly used method of moment can only handle a finite dimension of structure, the introduction of artificial edges on an infinite structure will result in spurious edge diffractions. Although the periodic-surface moment method was developed to improve this drawback, so far the finite conductivity of seawater cannot be well treated. In the traditional formulation, sear water are always regarded as a perfect conductor or modeled as a perfect conductor covered by a surface-impedance with complex number. In fact, the surface impedance is dependent on polarization, frequency and angle of incident wave and even is a tensor instead of scalar, therefore, this approach should have some problems in approximating realistic cases.

In this research, the sea wave is considered as a one-dimensionally periodic structure with complex dielectric constant. Under these assumptions, the artificial edge diffraction and finite conductivity of sea water can be modeled more realistic. Thus, not only the scattering characteristics but also the absorption by sea water could exactly realized. Since sea wave may has any arbitrary profiles of distribution, we have employed staircase approach to discretize them into rectangular ones, thus it will be changed into a multiple 1D periodic structures. Although the periodic variation is along one dimension, however, the scattering and guiding characteristics of periodic structures is inherently a three-dimensional boundary value problem that requires vector field representation to satisfy boundary condition at the interface.

In this research work, the method of mode matching will be employed to tackle the problem. There are two important steps: (1) the determination of a set of characteristics solutions for each constituent part including uniform and periodic regions, and (2) the imposing of boundary conditions requiring continuities of tangential electric and magnetic fields at each interface. Through the previous two steps, we could immediately determine the input-output relation of each layer containing periodic and uniform layers. The input-output relations of periodic or uniform layers can be considered as a building block and then the transmission line network representation will be employed to cascade them. Finally, the input-output relations as well as the scattering characteristics of overall structure could be easily determined by solving cascaded transmission line network. As we have described in the previous section, such a class of irregular shape of periodic structure should be taken into account to approach the realistic case. Based on the staircase approach, an irregular one can be partitioned into sufficiently thin layers as illustrated in attached figure, so each layer is characterized by a piece-wise constant permittivity. Thus a periodic layer with irregular profile can be considered as a cascade of rectangular shape periodic layers with the same period but different aspect ratio between constituent media.

The organization of this report is described as follows. The first part is the characteristic solutions of periodic medium, especially for the class of finite conductivity material, such as sea water. In this part, eigen-values and eigen-vectors of periodic waves, that is Floquet's solutions, in 1D periodic medium are derived rigorously. Since the relative dielectric constant involves a considerable conductivity, the concept of metal modes is taken into account to search the eigen-values for the dispersion relation. The second part is the electromagnetic boundary-value problem for the tangential electric and magnetic fields at the interfaces between adjacent constituent regions. Through this procedure, we could have the input-output relations of 1D periodic layer as well as uniform layers. In fact, these can be regarded as modules or building blocks for the overall structures. As long as these modules were well defined, we could deal with any arbitrary profiles of multiple 1D periodic structures with the same period. It is not necessary to reformulate the problem. That is the building block approach that will become clear later. The final part is the network representation of the problem. In this part, we will change the overall field problem into a cascade of transmission line network. And the basic transmission line circuit will be demonstrated to calculate the input-output relation of the overall structures. As it is done, the scattering characteristics of periodic sea wave will be totally determined.

2. Method of analysis

Characteristic solutions in one-dimensionally periodic medium

As we have described in previous, we would like to employ the mode-matching method to carry out the calculations of scattering characteristics of plane waves by 1D periodic structures. Before dealing with the boundary-value problem, we should have the characteristic solutions in periodic medium. According to the literatures there are two methods could be employed to deal with this problem: one is that of Fourier Method and the other is that of Modal Solution Method. The Fourier method is to utilize Fourier series to expand the periodic variation of periodic medium and also expand the general field solutions in such medium by using Floquet's solutions. As this is done, we could derive a dispersion relation of modes in such periodic medium. This dispersion roots usually could be considered as an eigenvalue problem and could be easily determined by commonly used subroutines. It is noted that the matrix is infinite in its size, however, in numerical computation, due to the finite source of memories we should truncate it into a fine one. In view of the finite truncation process, the eigenvalue of dispersion relation is no longer a rigorous one.

However, this method is not always valid for arbitrary number of relative dielectric constants. For example, if medium having high relative dielectric constant is considered, the Fourier series should include a large number of terms to ensure the convergence of Fourier expansion. It will lead to a convergence problem in solving the eigenvalue problem for a large size of matrix.

In this research work, we employed the later method, that is Modal Solution Method to deal with this problem. Here we consider a case of two-tone periodic medium as an example, however, for the other complicated case, this method is also available but with a slight modification.

As shown in attached figure, the periodic medium is infinite extension along x and z direction, and we assume there is no field and structure variation along y direction. The periodic medium having two relative dielectric constants ε1 and ε2, where the width of them are d1 and d2, respectively. In fact, the overall structure could be regarded as a multilayer structure with infinite number of layers. The characteristic of waves in such as a periodic structure can be analyzed by that of a unit cell imposing on a periodic boundary condition. This is a starting point of analysis for this problem.

The input-output relation of a finite thickness of uniform layer

For a periodic medium, the spatial variation of electric and magnetic field along z direction should be the same for each constituent region, which is $\exp(-jk_z z)$. However, for x direction, we could have a transmission line in each region. For each uniform transmission line, the input-output relation of a finite length of transmission line can be formulated in the transmission matrix or ABCD matrix, which could be given as:

$$\begin{pmatrix} v(x) \\ i(x) \end{pmatrix} = \begin{pmatrix} \cos k_x x & -jz_x \sin k_x x \\ -jy_x \sin k_x x & \cos k_x x \end{pmatrix} \begin{pmatrix} v(0) \\ i(0) \end{pmatrix} \qquad (1)$$

where kx is the propagation constant along x direction, zx (yx=1/zx) is the characteristic impedance (admittance) of the transmission line. The above equation describes the input-output relation of terminal voltage and current of a finite length transmission line.

As we have described in the previous, a two-tone periodic medium contains two different dielectric media in a unit cell. The input-output relation of such unit cell can be regarded as a cascade of two sections of transmission lines. Thus, we could have the ABCD matrix of the unit cell as follows:

$$\begin{pmatrix} v(d) \\ i(d) \end{pmatrix} = T_2(d_2)T_1(d_1)\begin{pmatrix} v(0) \\ i(0) \end{pmatrix} = T(d)\begin{pmatrix} v(0) \\ i(0) \end{pmatrix} \qquad (2)$$

$$T_2(d_2)=\begin{pmatrix} \cos k_x^{(2)}d_2 & -jz_x^{(2)}\sin k_x^{(2)}d_2 \\ -jy_x^{(2)}\sin k_x^{(2)}d_2 & \cos k_x^{(2)}d_2 \end{pmatrix} \quad (3)$$

$$T_1(d_1)=\begin{pmatrix} \cos k_x^{(1)}d_1 & -jz_x^{(1)}\sin k_x^{(1)}d_1 \\ -jy_x^{(1)}\sin k_x^{(1)}d_1 & \cos k_x^{(1)}d_1 \end{pmatrix} \quad (4)$$

Here, since we assume that the periodic medium is infinite extension along the periodicity, the relationship between the input and output terminals of a unit cell for the voltage and current waves just only differ by a phase shift for a periodic wave travels in a distance of one period, which is given as follows:

$$\begin{pmatrix} v(d) \\ i(d) \end{pmatrix} = e^{-jk_x d}\begin{pmatrix} v(0) \\ i(0) \end{pmatrix} \quad (5)$$

where k_x is the propagation constant of a periodic wave in such periodic medium. It is noted that the above equation can also be referred as periodic boundary condition in the literatures. Substitution of (5) into (2), we could have the dispersion relation of waves in periodic medium, which is equivalent to an eigenvalue problem in the following form:

$$T(d)\begin{pmatrix} v(0) \\ i(0) \end{pmatrix} = \lambda\begin{pmatrix} v(0) \\ i(0) \end{pmatrix} \quad (6)$$

therefore, the eigenvalue can be obtained by:

$$\det\left[T(d)-\lambda I\right]=0 \quad (7)$$

where I is the identity matrix, λ is the eigenvalue of the dispersion relation in(7), λ represents exp(-jkxd) and exp(+jkxd) for forward and backward propagating waves respectively. After some mathematical derivation, we could have the explicit form of the dispersion relation as follows:

$$\lambda^2-(T_{11}+T_{22})\lambda+(T_{11}T_{22}-T_{12}T_{21})=0 \quad (8)$$

the sum of the two eigenvalues could be expressed as trace of matrix T, that is:

$$\lambda_1+\lambda_2=T_{11}+T_{22}=2\cos k_x d \quad (9)$$

after substituting the elements of matrix T into (9), we could have the dispersion relation as follows:

$$\cos k_x^{(1)}d_1\cos k_x^{(2)}d_2-\frac{1}{2}\left(\frac{Z_x^{(1)}}{Z_x^{(2)}}+\frac{Z_x^{(2)}}{Z_x^{(1)}}\right)\sin k_x^{(1)}d_1\sin k_x^{(2)}d_2=\cos k_x d \quad (10)$$

where the characteristic impedance of transmission line of TE and TM waves in each region could be written as follows:

$$Z_x^{(i)} = \begin{cases} \dfrac{\omega\mu_o}{k_x^{(i)}}; \; TE \\ \dfrac{k_x^{(i)}}{\omega\varepsilon_o\varepsilon}; \; TM \end{cases} \tag{11}$$

$$k_x^{(i)} = \sqrt{k_o^2\varepsilon_i - k_t^2} \tag{12}$$

it is noted that the unknowns in (10) are kt and kx, thus we could look for kt for a given kx or look for kx for a given kt.

So far, we have derived the dispersion relation of waves in periodic medium. As long as we have the value of kx, we could determine the values of kt through a iteration procedure by using computer program. At the same time, we could also have the eigenvectors in (6), and then the distribution of current and voltage waves in such unit cell could be totally resolved.

Returning to (5), we could extend this equation to a more general form, that is:

$$\varphi(x+d) = e^{-jk_x d}\varphi(x) \tag{13}$$

where the scalar function φ represents either voltage or current waves. To satisfy the above equation, the scalar function φ should have the form follows:

$$\varphi(x) = e^{-jk_x x}\sum_{n=-\infty}^{+\infty} v_n e^{-jn\frac{2\pi}{d}x} \tag{14}$$

here, it could be easily proved as long as substitution of (14) into (13).

After having the voltage and current waves along x direction, we could determine the electric and magnetic field distributions in such periodic medium, and each tangential component in the (x,y) plane may be written as:

$$E_x'(x,y,z) = 0 \tag{15}$$

$$E_y'(x,y,z) = -\sum_m u_{zm}'\left[c_m' e^{-jk_{zm}''z} + d_m' e^{+jk_{zm}''z}\right]\sum_n V_n'^{(m)} e^{-jn\frac{2\pi}{d}x} \tag{16}$$

$$H_x'(x,y,z) = -\sum_m \left[c_m' e^{-jk_{zm}''z} - d_m' e^{+jk_{zm}''z}\right]\sum_n I_n'^{(m)} e^{-jn\frac{2\pi}{d}x} \tag{17}$$

$$H_y'(x,y,z) = -\sum_m u_{ym}'\left[c_m' e^{-jk_{zm}''z} - d_m' e^{+jk_{zm}''z}\right]\sum_n G_n'^{(m)} e^{-jn\frac{2\pi}{d}x}. \tag{18}$$

The above equations are those for TE^x modes, where the tangential components for TM^x modes are given as:

$$E_x''(x,z) = \sum_m \left[c_m'' e^{-jk_{zm}''z} + d_m'' e^{+jk_{zm}''z}\right]\sum_n V_n''^{(m)} e^{-jn\frac{2\pi}{d}x} \tag{19}$$

$$E''_y(x,z) = -\sum_m u''_{ym}\left[c''_m e^{-jk''_{zm}z} + d''_m e^{+jk''_{zm}z}\right]\sum_n G''^{(m)}_n e^{-jn\frac{2\pi}{d}x} \quad (20)$$

$$H''_x(x,y,z) = 0 \quad (21)$$

$$H''_y(x,z) = \sum_m u''_{zm}\left[c''_m e^{-jk''_{zm}z} - d''_m e^{+jk''_{zm}z}\right]\sum_n I''^{(m)}_n e^{-jn\frac{2\pi}{d}x} \quad (22)$$

In the above equations, the summation could be interchanged, thus we could expressed each equations in the matrix and vector form according to the order of Fourier components (or space harmonics), which are given as:

$$\underline{E}_x = \tilde{V}''\left[e^{-j\mathbf{k}''_z z}\underline{c}'' + e^{-j\mathbf{k}''_z z}\underline{d}''\right] \quad (23)$$

$$-\underline{E}_y = \tilde{V}'\left[e^{-j\mathbf{k}'_z z}\underline{c}' + e^{-j\mathbf{k}'_z z}\underline{d}'\right] + \tilde{G}''\left[e^{-j\mathbf{k}''_z z}\underline{c}'' + e^{-j\mathbf{k}''_z z}\underline{d}''\right] \quad (24)$$

$$\underline{H}_x = \tilde{I}'\left[e^{-j\mathbf{k}'_z z}\underline{c}' - e^{-j\mathbf{k}'_z z}\underline{d}'\right] \quad (25)$$

$$\underline{H}_y = -\tilde{G}'\left[e^{-j\mathbf{k}'_z z}\underline{c}' - e^{-j\mathbf{k}'_z z}\underline{d}'\right] + \tilde{I}''\left[e^{-j\mathbf{k}''_z z}\underline{c}'' - e^{-j\mathbf{k}''_z z}\underline{d}''\right] \quad (26)$$

where c's and d's are the vectors, which elements are the Fourier amplitudes corresponding to each space harmonic. k'_z and k''_z are the diagonal matrices of propagation constants along z direction for TE and TM modes, respectively. V, I and G are the coupling matrices which elements are similar to those from (16) to (22) but with coordinate transformations, which are given as:

$$\tilde{V}' = \left(u'_{zm}V'^{(m)}_n\right) \quad (27)$$

$$\tilde{I}' = \left(I'^{(m)}_n\right) \quad (28)$$

$$\tilde{G}' = \left(u'_{ym}G'^{(m)}_n\right) \quad (29)$$

$$\tilde{V}'' = \left(V''^{(m)}_n\right) \quad (30)$$

$$\tilde{I}'' = \left(u''_{zm}I''^{(m)}_n\right) \quad (31)$$

$$\tilde{G}'' = \left(u''_{ym}G''^{(m)}_n\right) \quad (32)$$

Here, we could express the tangential electric and magnetic fields in terms of super-matrix and vector forms, which are given as:

$$\underline{E}_t = \mathbf{P}\left(e^{-j\mathbf{k}_z z}\underline{c} + e^{+j\mathbf{k}_z z}\underline{d}\right) \quad (33)$$

$$\underline{H}_t = \mathbf{Q}\left(e^{-j\mathbf{k}_z z}\underline{c} - e^{+j\mathbf{k}_z z}\underline{d}\right) \quad (34)$$

where matrix c and d are the forward and backward propagation vector with their elements represent the Fourier amplitude of each space harmonic, including TE and TM modes. Where the coupling matrices P and Q are given as:

$$\mathbf{P}=\begin{pmatrix}\tilde{V}' & \tilde{G}''\\ O & \tilde{V}''\end{pmatrix} \quad (35)$$

$$\mathbf{Q}=\begin{pmatrix}\tilde{I}' & O\\ -\tilde{G}' & \tilde{I}''\end{pmatrix} \quad (36)$$

<u>Input-output relation of 1D periodic layer</u>

So far, we have derived the general solutions of electric and magnetic fields in the 1D periodic medium. The tangential electric and magnetic fields including TE and TM modes are also expressed in a super- matrix and vector forms. For a finite thickness of 1D periodic structure, the input-output relation of electric and magnetic fields can be obtained as the termination conditions is determined, which can be given as:

$$\underline{E}_t(0)=\mathbf{Z}_{in}\underline{H}_t(0) \quad (37)$$

$$\mathbf{Z}_{in}=\mathbf{P}\left(\mathbf{I}+\mathbf{\Gamma}_l\right)\left(\mathbf{I}-\mathbf{\Gamma}_l\right)^{-1}\mathbf{Q}^{-1} \quad (38)$$

$$\mathbf{\Gamma}_l=e^{-j\mathbf{k}_z t}\mathbf{\Gamma}_{out}e^{-j\mathbf{k}_z t} \quad (39)$$

$$\mathbf{\Gamma}_{out}=\left(\mathbf{Z}_{out}\mathbf{Q}+\mathbf{P}\right)^{-1}\left(\mathbf{Z}_{out}\mathbf{Q}-\mathbf{P}\right) \quad (40)$$

$$\underline{E}_t(t)=\mathbf{Z}_{out}\underline{H}_t(t) \quad (41)$$

$$\underline{E}_t(t)=\mathbf{P}\left(\mathbf{I}+\mathbf{\Gamma}_{out}\right)e^{-j\mathbf{k}_z d}\left(\mathbf{I}+\mathbf{\Gamma}_l\right)^{-1}\mathbf{P}^{-1}\underline{E}_t(t) \quad (42)$$

3. Numerical Results and Discussions

In this research work, based on the theoretical formulation in previous section, we have developed a program to analyze the scattering characteristics of plane wave by an arbitrary profile of 1D periodic structure. Here, an arbitrary profile of 1D periodic structure could have spatial variation along the direction perpendicular to the periodicity. We have utilized staircase approach to model it as a multiple 1D periodic layers with rectangular profile. On the other hand, to approach the realistic case of sea-water, the relative dielectric constant of sea-water is assumed to be 23-j32 (in 20° C). The frequency of incident plane wave is 1GHz.

To realize the basic physical picture of waves scattering by sea-wave modeling by periodic structure, we have modeled sea-wave by a rectangular profile of periodic structure with the aspect ratio between sea and air is 0.5. The period of such periodic structure is assumed to be 600cm, where the height of it is varied from 5 cm to 80 cm in our simulation to observe the physical meanings.

Before doing extensive numerical simulations, we have carried out the convergence test for this problem. In principle, there is no limitation of the period, however, based on the basic characteristics of periodic structures, we know that the number of propagation modes will increase as the increasing of the ratio between period and wavelength. In our case, the ration between period and wavelength is up to 20, that is, we should have considerable number of propagation modes to ensure the convergence for the numerical computations. Figure 1 is the variation of diffraction efficiency of space harmonic n = 0 versus number of space harmonic (N) that we have used, for TE and TM polarization both. It is noted that the indices of space harmonic used is from –N to +N; that is, the total number should be 2N+1. The height of such periodic structure is assumed to be 80 cm. The plane wave is normal incidence with the frequency 1GHz. From the results in figure 2, we have observed that the convergence rate is excellent for both TE (in solid line) and TM (in dash line) polarizations. It is noted that as the number of space harmonic is greater than 20, the variation of result is less than 1%. To obtain a accurate simulation result, we have utilized 101 space harmonics for the ensuing numerical examples.

Figure 3 is the variation of reflection efficiency of space harmonic n = 0 versus incident angle for various height of sea-wave. We have changed the height from 5 cm to 80 cm to see the variation of reflectivity for space harmonic n = 0. It is noted that this space harmonic (n = 0) is reflected return to the incident direction, which is similar to the mono-static RCS in radar detection. We have observed that the diffraction efficiency will have large variation as the increase of the height. Based on geometric optics, it may be concluded that as the increasing of height, the multiple reflections will dominant; therefore, the diffraction efficiency will experience a large variation for the changes of incident angle. On the contrary, the result of the case of smallest height has no damping, which approaches the case of scattering by flat plane.

As depicted in Figure 4, we have compared the reflectivity of sea-wave by TE and TM incident plane waves. In general, we have found that these two polarizations of incident plane waves share the similar reflection characteristics. However, the reflectivity of TE wave is larger than that of TM one. Since the rectangular profile of sea-wave is similar to a parallel-plate waveguide, meanwhile, the real part and imaginary part of relative dielectric constant are not small, thus the field hard to penetrate into such medium. The wave in air region is similar to that of parallel-plate waveguide modes. As we have known, the lowest TM mode (TM_0) can exist in PPWG without cutoff frequency but the TE is not. This will result in the reflection of TE mode is obvious than that of TM mode, as shown in this figure. On the other hand, it is noted that since the incident of TM plane waves can have the characteristics of total transmission, the reflectivity will vanish at certain incident angle. In this case, the total transmission angle will be given approximated as:

$$\theta = \tan^{-1}\frac{\sqrt{\varepsilon_2}}{\sqrt{\varepsilon_1}} \approx \tan^{-1}\sqrt{\frac{23}{1}} \approx 78^o . \qquad (43)$$

As shown in this figure, the total transmission angle is located around such angle as expected in the above equation. It is noted that the reflectivity becomes a large one at large incident angle (low grazing incident angle) for both case of TE and TM polarizations, as shown in this figure.

Figure 5 indicates the absorption efficiency of plane waves by such sea-wave versus incident angle. As we have described in the previous, the TM waves can propagate through the PPWG without cutoff but the TE wave can't. Thus, the absorption of incident plane wave will be obvious for TM wave than that of TE one.

4. Conclusions

In this research work, we have modeled sea-wave by a one-dimensionally periodic structure. We have repeated the waveform of a sea-wave to a periodic structure of infinite extension. As long as the period is large enough, this will approach the realistic case but without the artificial edge diffraction obtained by the other methods proposed in the literatures. Moreover, since the complex relative dielectric constant is utilized in this research work, the absorption of plane waves by sea-water could also be figured out correctly. We have employed the rigorous mode-matching method to analyze this problem and extensive numerical results have been shown and explained in the previous section. Since the profile of sea-wave is changed randomly, it is important to analyze the scattering characteristics by statistical method. However, this research work has build up the bases for numerical computations for an arbitrary profile of sea-wave.

Appendix

Since the structure is uniform in (y,z) plane, there are two uncoupled complete sets denoted by TE^x and TM^x. The field components of these two sets are given as:

TE^x

$$\underline{F} = \underline{x}_o \psi(x,y,z) \tag{44}$$

where the scalar function ψ satisfies the wave equation as follows:

$$\left(\nabla^2 + k_o^2 \varepsilon_i\right)\psi = 0 \tag{45}$$

$$\underline{E} = -\nabla \times \underline{F} \tag{46}$$

after the electric field is obtained by the cross product operation in the above equation, we could determine magnetic field by Maxwell equation.

$$\nabla \times \underline{E} = -j\omega\mu_o \underline{H} \tag{47}$$

Since we have assumed that there are no variations on the field as well as structures along y direction, the scalar function is independent of y and the scalar function can be defined as follows:

$$\psi(x,z) = e^{-jk_z x}\phi(x) \tag{48}$$

TEx	TMx
$E'_x(x,z) = 0$	$E''_x(x,z) = \dfrac{k''_t}{\omega\varepsilon_o\varepsilon(x)} i''(x)$
$E'_y(x,z) = -\dfrac{k'_z}{k'_t} v'(x)$	$E''_y(x,z) = -\dfrac{k_y}{k''_t} v''(x)$
$E'_z(x,z) = \dfrac{k'_y}{k'_t} v'(x)$	$E''_z(x,z) = -\dfrac{k''_z}{k''_t} v''(x)$
$H'_x(x,z) = -\dfrac{k'_t}{\omega\mu_o} v'(x)$	$H''_x(x,z) = 0$

$H'_y(x,z) = -\frac{k_y}{k'_t} i'(x)$	$H''_y(x,z) = \frac{k''_z}{k''_t} i''(x)$
$H'_z(x,z) = -\frac{k'_z}{k'_t} i'(x)$	$H''_z(x,z) = -\frac{k_y}{k''_t} i''(x)$

Figure captions:

Figure 1: Structural configuration of a sea-wave which is modeled by a multiple 1D periodic layers.

Figure 2: Convergence test for reflectivity of space harmonic n = 0 versus number of space harmonic. The dash and solid line are that for TE and TM polarizations, respectively.

Figure 3: Variation of normalized reflectivity efficiency of space harmonic n = 0 versus incident angle for various height of sea-wave: TE polarization.

Figure 4: Comparison of reflectivity efficiency for TE and TM incident plane waves.

Figure 5: Absorption of incident plane waves by sea-wave: TE and TM polarizations.



Figure 1: Structural configuration of a sea-wave which is modeled by a multiple 1D periodic layers.



Figure 2: Convergence test for reflectivity of space harmonic n = 0 versus number of space harmonic. The dash and solid line are that for TE and TM polarizations, respectively.



Variation of normalized reflected power of space harmonic n=0 versus incident angle for various height of sea-wave



Figure 4: Comparison of reflectivity efficiency for TE and TM incident plane waves.



Figure 5: Absorption of incident plane waves by sea-wave: TE and TM polarizations.

```fortran
        PROGRAM MAIN

!  Research Program: Plane wave scattering by stack of 1D periodic layers
!  TE and TM polarization both
!  IC=1 TE polarization and IC=2 for TM polarization
!  This theory and code are Developed and Edited by
!              Ruey Bing Hwang, Ph.D
!  If you have any questions, please to forward your message to:
!  rbhwang@eic.nctu.edu.tw

   IMPLICIT NONE
   Character*1 :: QUE
   INTEGER     :: I,NM,NP,NT,NLAYERS,IC,ITSH
   Integer     :: NSC,Nstart,Nstop,Ninc
   Integer     :: WHPM,PNO,N1,N2,N3,N4,N5,NS1,NS2,NS3,NS4,NS5,OSH
   Real        :: Pmax,Pmin,IPP(2)
   Real        :: THETA,XLAMDA
   Complex     :: Es,Ea
   REAL,allocatable,dimension(:)    :: D1,D2,H,DL,PRX,PTX,SP
   Complex,allocatable,dimension(:) :: XER1,XER2
   Common /INOUT/ Ea,Es
   COMMON /GIVEN/ THETA,XLAMDA
   COMMON /MODES/ NM,NP

   OPEN(UNIT=1,FILE='INPUT.DAT')
   OPEN(UNIT=2,FILE='RXSH.DAT')
   OPEN(UNIT=3,FILE='TXSH.DAT')
   OPEN(UNIT=5,FILE='OUTPUT.DAT ')
   OPEN(UNIT=7,FILE='CONV.DAT')

   Read(1,*)
   READ(1,*) IC   ! Polarization of incident plane wave
   Read(1,*)
   READ(1,*) Ea   ! Relative Diel. const in incident region
   Read(1,*)
   Read(1,*) Es   ! the Complex diel. constant in output region
   Read(1,*)
   READ(1,*) IPP(1)   ! XLAMDA: Iteratable
   Read(1,*)
   Read(1,*) IPP(2)   ! THETA: Iteratable
   Read(1,*)
   Read(1,*) N1,N2,N3,N4,N5  ! the indices of space harmonics to be saved
   Read(1,*)
   Read(1,*) NLAYERS  ! the number of layers

   Allocate(XER1(NLAYERS))
    Allocate(XER2(NLAYERS))
     Allocate(D1(NLAYERS))
      Allocate(D2(NLAYERS))
       Allocate(H(NLAYERS))
        Allocate(DL(NLAYERS))
```



```fortran
   Do I=1,NLAYERS
     Read(1,*) XER1(I),XER2(I),D1(I),D2(I),H(I),DL(I)
   End Do

!  WRITE(*,*)
!  WRITE(*,*) '>>> Pls. input the index number of space harmonic of excitation =?'
!  Read (*,*)      ITSH
!  WRITE(*,*)

   ITSH=0


   Call Title

!  ----- Part 1: Convergence test ------------------------------------

   Write(*,*) '>>> Do you want to perform a convergence test (y or n) ???'
   Read (*,*)      QUE
   Write(*,*)

   If ((QUE .EQ. 'y') .or. (QUE .EQ. 'Y')) then

   Write(*,*) '>>> Input the start/stop/inc number of space harmonics to '
   Write(*,*) '    take a convergence test !!!'
   Read(*,*)       Nstart,Nstop,Ninc

   NSC=Nstart
   Do while ((NSC .LE. Nstop) .and. (NSC .GE. Nstart))
     NP=NSC
      NM=-NSC
       NT=2*NSC+1
        Allocate(PRX(NT))
        Allocate(PTX(NT))
         XLAMDA=IPP(1)
          THETA=IPP(2)
           Call SCATT(PRX,PTX,NT,IC,ITSH,NLAYERS,XER1,XER2,D1,D2,H,DL)
            Write(*,*) NSC,PRX(NSC+1),PRX(NSC+2),SUM(PRX+PTX)
            Write(7,*) NSC,PRX(NSC+1),PRX(NSC+2),SUM(PRX+PTX)
             NSC=NSC+Ninc
              Deallocate(PRX,PTX)
   End Do
   Else
   End If
   Close(7)

!  -------- Iteration of the parameters ------------------------------
    CALL Menu(Pmax,Pmin,PNO,whpm)
      ALLOCATE(SP(PNO))
       CALL scan(SP,Pmax,Pmin,PNO)

!  --- Input the Max. positive index of space harmonic  ---
       Write(*,*)
```

```
      Write(*,*) '>>> Input the Positive and negative'
      Write(*,*) '>>> indices of space harmonics (two integers) ?'
      Write(*,*)
      Read(*,*) NP,NM

    IF     (WHPM .eq. 1) THEN
           Write(*,*) " >> Wavelength Iteration !!! "
           WRITE(2,*) " >> Reflection Power VS. Wavelength "
           Write(2,*)
           Write(5,*) " >> Wavelength Iteration !!! "
           Write(5,*) " >> Incident Angle=",IPP(2)
           Write(5,*)

           IF (IC .EQ. 1) then
           Write(2,*) '>>> TE polarization'
           Write(5,*) '>>> TE polarization'
           Else
           Write(2,*) '>>> TM polarization'
           Write(5,*) '>>> TM polarization'
           End If

           Write(2,*) '>>> The incident angle is',IPP(2)
           Write(2,*) '>>> The indices of the space harmonics are',NP,NM
           Write(2,*)

           WRITE(3,*) " >> Transmission Power VS. Wavelength "

           IF (IC .EQ. 1) then
           Write(3,*) '>>> TE polarization'
           Else
           Write(3,*) '>>> TM polarization'
           End If
           Write(3,*) '>>> The relative diel. constant of the input and output region
s are',Ea,Es
           Write(3,*) '>>> The incident angle is',IPP(2)
           Write(3,*) '>>> The indices of the space harmonics are',NP,NM
           Write(3,*)


    ELSEIF (WHPM .eq. 2) THEN
           Write(*,*) " >> Incident Angle Iteration !!! "
           WRITE(2,*) " >> Reflection Power VS. Incident Angle "

           Write(5,*) " >> Incident Angle Iteration !!! "
           Write(5,*) " >> Wavelength=",IPP(1)
           WRITE(5,*) " >> Reflection Power VS. Incident Angle "

           IF (IC .EQ. 1) then
           Write(2,*) '>>> TE polarization'
           Write(5,*) '>>> TE polarization'
           Else
           Write(2,*) '>>> TM polarization'
```



```
           Write(5,*) '>>> TM polarization'
           End If

           WRITE(3,*) " >> Transmission Power VS. Incident Angle "
           IF (IC .EQ. 1) then
           Write(3,*) '>>> TE polarization'
           Else
           Write(3,*) '>>> TM polarization'
           End If

    END IF

! >>> Selecting space harmonics to observe !!!
      NS1=OSH(N1,NP)
       NS2=OSH(N2,NP)
        NS3=OSH(N3,NP)
         NS4=OSH(N4,NP)
          NS5=OSH(N5,NP)
         NT=NP-NM+1
         Allocate(PRX(NT))
          Allocate(PTX(NT))

           Write(*,*) '>>> The indices of space harmonics are from',NP,'to',NM
           Write(*,*) '>>> The total number of space harmonics is',NT

    Write(5,*)
    Write(5,*) " -------------------------------------------------------"

    Do I=1,Pno
        write(*,*) " >> Pass #=",I
       IPP(whpm)=SP(I)
        XLAMDA=IPP(1)
         THETA=IPP(2)
        Call SCATT(PRX,PTX,NT,IC,ITSH,NLAYERS,XER1,XER2,D1,D2,H,DL)
             Write(*,1) SP(I),SUM(PRX),SUM(PTX),SUM(PRX)+SUM(PTX)
             Write(5,1) SP(I),SUM(PRX),SUM(PTX),SUM(PRX)+SUM(PTX)
             Write(2,1) SP(I),PRX(NS1),PRX(NS2),PRX(NS3),PRX(NS4),PRX(NS5)
             Write(3,1) SP(I),PTX(NS1),PTX(NS2),PTX(NS3),PTX(NS4),PTX(NS5)
1            FORMAT(F8.4,5(1X,F8.6))
    End do

   Deallocate(SP,PRX,PTX)

  END


! -------------------------------------------------------------------
  Subroutine SCATT(PRX,PTX,NT,IC,ITSH,NLAYERS,XER1,XER2,D1,D2,H,DL)
! -------------------------------------------------------------------
  Use MSIMSL
```

```fortran
 Implicit None
 Integer :: NP,NM,NT,NO,I,n,NLAYERS,IC,ITSH
 Real    :: H(NLAYERS),D1(NLAYERS),D2(NLAYERS),DL(NLAYERS)
 Real    :: DX,Ko,PRX(NT),PTX(NT),PINC,TWOPI=2.0*3.141592654
 Real    :: THETA,XLAMDA
 Complex :: Ea,Es,XER1(NLAYERS),XER2(NLAYERS),ZKx,ZKxn,Zkzn,Zna,Zns
 Complex :: CZERO=(0.0,0.0)
 Complex,Dimension(NT)    :: VR,VT,VINC,YAV,YSV
 Complex,Dimension(NT,NT) :: ZA,ZS,ZIN,ZOUT,TXX,TXP,INVXX,REFL,U
! External   FSHIFT
 Common /INOUT/ Ea,Es
 COMMON /GIVEN/ THETA,XLAMDA
 COMMON /MODES/ NM,NP

 DX=D1(1)+D2(1)
 Ko=TWOPI/XLAMDA
 ZKX=Ko*CSQRT(Ea)*SIND(THETA)
 Call EYE(U,NT)

! >>> Characteristic Impedance of space harmonics in the input
!     and output regions, respectively

     ZA=CZERO
      ZS=CZERO
       Do I=1,NT
           n=NP-I+1
            ZKxn=ZKx+(n*TWOPI/DX)
             Call IMPEDANCE(Zna,Ea,ZKxn,ZKzn,ko,IC)
              Call IMPEDANCE(Zns,Es,ZKxn,ZKzn,ko,IC)
               ZA(I,I)=Zna
                YAV(I)=1.0/Zna
               ZS(I,I)=Zns
                YSV(I)=1.0/Zns
       End Do

! >>> Entering the NPL-th periodic layer
     ZOUT=ZS
     TXX=U


     Do I=NLAYERS,1,-1
        Call XSLAYER(ZIN,TXP,NT,XER1(I),XER2(I),D1(I),D2(I),H(I),DL(I),ZKX,XLAMDA,ZOUT
,IC)
         TXX=MATMUL(TXX,TXP)
          ZOUT=ZIN
     End Do


! >>> Calculation of the reflection matrix
         CALL LINCG(NT,ZIN+ZA,NT,INVXX,NT)
          REFL=Matmul(ZIN-ZA,INVXX)

!   >>>  Incident plane wave is the space harmonic of ITSH-th space harmonic
```



```fortran
!   >>>  at an incident angle theta
!   >>>  REFLECTED Power
         NO=NP-ITSH+1
         VR=REFL(:,NO)
         PINC=REAL(YAV(NO))
         PRX=REAL(CABS(VR*VR)*YAV)/PINC

!   >>>  Transmitted Power
         VINC=VR
         VINC(NO)=VR(NO)+1.0
         VT=MATMUL(TXX,VINC)
         PTX=REAL(CABS(VT*VT)*YSV)/PINC


         Return
         End

! ---------------------------------------------------------------
  Subroutine XSLAYER(ZIN,TX,NT,ER1,ER2,D1,D2,H,DL,ZKX,LAMDA,ZOUT,IC)
! ---------------------------------------------------------------
  IMPLICIT NONE
  Integer :: NT,IC
  Complex :: ZIN(NT,NT),TX(NT,NT),ZOUT(NT,NT),ER1,ER2,ZKX
  Real    :: D1,D2,H,DL,LAMDA,ER

  ER=CABS(ER1-ER2)

  If (ER .LE. 1.e-6) then
    Call SULAYER(ZIN,TX,H,D1+D2,ER1,NT,ZKX,ZOUT,IC)
  Else
    Call SPLAYER(ZIN,TX,NT,ER1,ER2,D1,D2,H,DL,ZKX,LAMDA,ZOUT,IC)
  End If
  Return
  End
! ---------------------------------------------------------------
  Subroutine SPLAYER(ZIN,TX,NT,ER1,ER2,D1,D2,H,PS,ZKX,XLAMDA,ZOUT,IC)
! ---------------------------------------------------------------
! Thius subroutine calculate the input-output relation
! of a single periodic layer.
! In this subroutine, the Fourier series expansion method
! was employed to expand the complex dielectric constant
!
! Parameter descreption:
! Period: DX=D1+D2
! IC=1 : TE polarization
! IC=2 : TM polarization
  USE MSIMSL
  IMPLICIT NONE
  INTEGER  :: NT,NM,NP,I,K,J,n,IC
  REAL     :: DX,XLAMDA,PS,Ko,H,D1,D2,sko
  REAL     :: TWOPI=2.0*3.141592654
  COMPLEX  :: FC,FCI,TEMP,PHX
```

```
 Complex   :: ZKx,CC,C
 Complex   :: ER1,ER2,Zj=(0.0,1.0),CZERO=(0.0,0.0)
 COMPLEX,dimension(NT)     :: KAPA
 Complex,dimension(NT,NT) :: D,IVD,DB,IVDB,PMAT,QMAT,XX,INVXX,PHASE,ZGC,YGC,  &
                             ZOUT,REFLD,REFGS,ZIN,U,TX,OMAG,KXMAT
 External FC,FCI
 COMMON /MODES/ NM,NP

 DX=D1+D2
 Ko=TWOPI/XLAMDA
 sko=Ko*Ko

! >>> ------   Generating the Fourier Coefficient Matrix ----- <<<

 D=CZERO        !    E(x)
  IVD=CZERO     ! Inv(E(x))
   DB=CZERO     !     1/E(x)
    IVDB=CZERO ! INV(1/E(x))

       Call EYE(U,NT)

      KXMAT=CZERO

      DO I=1,NT
         n=NP-I+1
          KXMAT(I,I)=ZKx+(n*TWOPI/DX)
      End Do

      Do I=1,NT
          Do J=1,NT
             K=I-J
              D(I,J)=FC(K,ER1,ER2,D1,D2,PS)
               DB(I,J)=FC(K,1.0/ER1,1.0/ER2,D1,D2,PS)
          End Do
       End Do

       Call LINCG(NT,D,NT,IVD,NT)
        Call LINCG(NT,DB,NT,IVDB,NT)
! >>> -------------------------------------------------------- <<<


! >>>
     Select Case (IC)

     Case(1)     ! >>> TE polarization
       OMAG=sko*D-MATMUL(KXMAT,KXMAT)
! >>>   Find eigenvalues and vectors of A
        CALL EVCCG (NT,OMAG,NT,KAPA,QMAT,NT)
!    >>> The z-direction propagation constant of each space harmonics
!        in periodic medium

        PHASE=CZERO
```



```
        YGC=CZERO

        Do I=1,NT
           CC= KAPA(I)
            Call XSIGN(CC,C)
              YGC(I,I)=C/Ko
               TEMP=-Zj*C*H
               IF (REAL(TEMP) .LT. -40.0) then
                   PHX=0.0
               Else
                   PHX=CEXP(TEMP)
               End If
               PHASE(I,I)=PHX
        End Do

        PMAT=MATMUL(QMAT,YGC)


     Case (2)   ! >>> TM polarization

        OMAG=Matmul(IVDB,sko*U-MATMUL(KXMAT,MATMUL(IVD,KXMAT)))

!    >>> Find eigenvalues and vectors of A
        CALL EVCCG (NT,OMAG,NT,KAPA,PMAT,NT)

!    >>> The z-direction propagation constant of each space harmonics
!        in periodic medium

        PHASE=CZERO
        ZGC=CZERO

        Do I=1,NT
           CC= KAPA(I)
            Call XSIGN(CC,C)
              ZGC(I,I)=C/Ko
               TEMP=-Zj*C*H
               IF (REAL(TEMP) .LT. -40.0) then
                   PHX=0.0
               Else
                   PHX=CEXP(TEMP)
               End If
               PHASE(I,I)=PHX
        End Do

        QMAT=MATMUL(DB,MATMUL(PMAT,ZGC))

     Case Default
     End Select


! >>> Input-Output Relation of the single periodic layer
```

```
        XX=MATMUL(ZOUT,PMAT)+QMAT
         Call LINCG(NT,XX,NT,INVXX,NT)
          XX=MATMUL(ZOUT,PMAT)-QMAT
           REFGS=MATMUL(INVXX,XX)         ! STORED
            REFLD=MATMUL(PHASE,MATMUL(REFGS,PHASE))

        XX=MATMUL(PMAT,U-REFLD)
         Call LINCG(NT,XX,NT,INVXX,NT)
          XX=MATMUL(QMAT,U+REFLD)
           ZIN=MATMUL(XX,INVXX)

! >>>   Transfer Matrix through this periodic layer
          XX=MATMUL(QMAT,U+REFLD)
           Call LINCG(NT,XX,NT,INVXX,NT)
            TX=MATMUL(MATMUL(QMAT,U+REFGS),MATMUL(PHASE,INVXX))

  RETURN
  END
! ------------------------------------------------------------
  Subroutine SULAYER(ZIN,TX,H,DX,ER,NT,ZKX,ZOUT,IC)
! ------------------------------------------------------------
! Thius subroutine calculate the input-output relation
! of a single periodic layer.
! In this subroutine, the Fourier series expansion method
! was employed to expand the complex dielectric constant
!
! Parameter descreption:
! Period: DX=D1+D2

  USE MSIMSL
  IMPLICIT NONE
  INTEGER  :: NT,NM,NP,I,n,IC
  REAL     :: THETA,XLAMDA,Ko,H,DX
  REAL     :: TWOPI=2.0*3.141592654
  COMPLEX  :: ER,TEMP,PHX
  Complex  :: ZKx,ZKxn,Zn,ZKzn
  Complex  :: Zj=(0.0,1.0),CZERO=(0.0,0.0)
  Complex,dimension(NT,NT) :: XX,INVXX,PHASE,ZOUT,REFLD,REFGS,ZIN,U,TX,ZC,YC
  COMMON /GIVEN/ THETA,XLAMDA
  COMMON /MODES/ NM,NP

  Ko=TWOPI/XLAMDA

  Call EYE(U,NT)

  YC=CZERO
  ZC=CZERO
  PHASE=CZERO

    Do I=1,NT
       n=NP-I+1
```



```
     ZKxn=ZKx+(n*TWOPI/DX)
      Call IMPEDANCE(Zn,ER,ZKxn,ZKzn,ko,IC)
         YC(I,I)=1.0/Zn
         ZC(I,I)=Zn
          TEMP=-Zj*ZKzn*H
           IF (REAL(TEMP) .LT. -40.0) then
               PHX=0.0
           Else
               PHX=CEXP(TEMP)
           End If
               PHASE(I,I)=PHX
  End Do

      XX=MATMUL(ZOUT,YC)+U
       Call LINCG(NT,XX,NT,INVXX,NT)
        XX=MATMUL(ZOUT,YC)-U
         REFGS=MATMUL(INVXX,XX)         ! STORED
          REFLD=MATMUL(PHASE,MATMUL(REFGS,PHASE))

      XX=MATMUL(YC,U-REFLD)
       Call LINCG(NT,XX,NT,INVXX,NT)
         ZIN=MATMUL(U+REFLD,INVXX)

! >>>   Transfer Matrix through this uniform layer
        XX=U+REFLD
         Call LINCG(NT,XX,NT,INVXX,NT)
          TX=MATMUL(U+REFGS,MATMUL(PHASE,INVXX))

  RETURN
  END
! ---------------------------------------------------
  Complex Function FC(m,E1,E2,D1,D2,PS)
! ---------------------------------------------------
! Fourier Series Expansion of ER(x) for the periodic
! Structures
! PS: shift of the distance in the lateral direction

  Implicit none
  Integer :: m
  Complex :: E1,E2,Zj=(0.0,1.0),PHASE
  Real    :: PS,D1,D2,DX,DCY,PI=3.141592654
  DX=D1+D2
  PHASE=CEXP(Zj*m*2.0*PI*PS/DX)
  DCY=D1/(D1+D2)
  If (m .EQ. 0) then
     FC=DCY*E1+E2*(1.0-DCY)
  Else
     FC=PHASE*(E1-E2)*SIN(m*PI*DCY)/m/PI
  End If
  Return
  End
!-----------------------------------------------------------
```

```
  SUBROUTINE XSIGN(CC,C)
!---------------------------------------------------------------
  COMPLEX CC,C
!---------------------------------------------------------------
! DETERMINING THE SIGN OF A SQUARE ROOT FOR THE TRANSVERSE
! WAVENUMBER, WHEN THE LONGETUDINAL WAVENUMBER IS KNOWN
! DECIDING THE PHASE CONSTANT OF A FORWARD PROPAGATING MODE
!---------------------------------------------------------------
! Input to this subroutine is CC = sko*Er+sa-ab + j*(2*a*b-sko*Ei),
! where sko=ko*ko, sa=a*a, sb=b*b, Er=Re(E), and Ei=Im(E)
! The output is C=sqrt(CC)
! If ARGR=Re(CC) is positive, the imagineary part of C will have the
! same sign as that of CC  ---  accept the convention of computers
! If ARGR <= 0, the wave must be decaying in the transvers direction
! Im(C) < 0.  Therefore, when ARGI=Im(CC) > 0, we must choose the
! negative sign for the square root
!  ***  These criteria apply to leaky wave structures   ***
!  ***  as well as lossy waveguides                     ***
  ARGR = REAL(CC)
  ARGI = AIMAG(CC)
  Call CSR(CC,C)
  IF(ARGR .LE. 0.0  .AND. ARGI .GE. 0.0) C = -C
  RETURN
  END
!---------------------------------------------------------------
  Subroutine CSR(S,C)
!---------------------------------------------------------------
  Complex S,C
  Real    P,A
  A=CABS(S)
  P=ATAN2(AIMAG(S),REAL(S))/2.0
  C=SQRT(A)*CMPLX(cos(P),sin(P))
  Return
  End
!---------------------------------------------------------------
  Subroutine EYE(U,NN)
!---------------------------------------------------------------
  Implicit none
  Integer :: I,NN
  Complex :: U(NN,NN)
  U=(0.0,0.0)
  Do I=1,NN
    U(I,I)=1.0
  End Do
  Return

!---------------------------------------------------------------
  Subroutine IMPEDANCE(Z,ER,ZKx,ZKz,ko,mode)
!---------------------------------------------------------------
  Implicit None
  Integer :: mode
  Real    :: ko,sko
```



```
  Complex :: Z,SKz,ZKz,ZKx,ER
  sko=ko*ko
  SKz=sko*ER-ZKx*ZKx
  Call XSIGN(SKz,ZKz)
  Select Case (mode)
  Case(1)
   Z=Ko/ZKz
  Case(2)
   Z=ZKz/ER/Ko
  Case Default
  end select
  Return
  End
!---------------------------------------------------------------
  Subroutine Menu(Pmax,Pmin,Pno,whpm)
!---------------------------------------------------------------
  Implicit none
  Integer :: Pno,whpm
  Real    :: Pmin,Pmax

  Write(*,*)
  Write(*,*)
  Write(*,*)
  Write(*,*) " >> There are TWO parameters could be changed, "
  Write(*,*) "    You can select one of them to iterate !!!: "
  Write(*,*)
  Write(*,*)
  Write(*,*) "    1) Wavelength of Incidence Wave  "
  Write(*,*)
  Write(*,*) "    2) Incident angle; in Degree "
  Write(*,*)
  Write(*,*) "    3) QUIT !!! "
  Write(*,*)
  Write(*,*)
  Write(*,*) "    Which parameter do you want to iterate ? "
   Read(*,*)     whpm

  If (whpm .NE. 3) then
  WRITE(*,*)
  Write(*,*) " >> Please input the 1. minimume Value "
  Write(*,*) "                     2. Maximume Value "
  Write(*,*) "                     3. Number of points"
   Read(*,*)     Pmin,Pmax,Pno
  Write(*,*)     Pmin,Pmax,Pno
  write(*,*)
  Else
     STOP
  End If
  Return
  End
!---------------------------------------------------------------
  Subroutine scan(SP,Pmax,Pmin,Pno)
```

```
!--------------------------------------------------------------------
  Implicit none
  Integer :: Pno,I
  Real    :: SP(Pno),Pinc,Pmax,Pmin

  If (Pno .EQ. 1) then
      SP(1)=Pmin
  Else
      Pinc=(Pmax-Pmin)/(Pno-1)
      Do I=1,Pno
         SP(I)=Pmin+(I-1)*Pinc
      End do
  End if
  Return
  End
!--------------------------------------------------------------------
  Integer Function OSH(N,NP)
!--------------------------------------------------------------------
  Implicit none
  Integer :: N,NP
  OSH=NP-N+1
  Return
  End

! -------------------------------------------------------------------------------
   Subroutine Title
! -------------------------------------------------------------------------------
   Write(*,*)
   WRITE(*,*) '     ============================================================'
   WRITE(*,*) '     *  Research Program: Plane Wave Scattering by Stacks of     *'
   WRITE(*,*) '     *                    Periodic Layers                        *'
   WRITE(*,*) '     *                                                           *'
   WRITE(*,*) '     *  This theory and code are Developed and Edited by         *'
   WRITE(*,*) '     *   Ruey Bing Hwang, Ph.D    August 8th, 2000               *'
   WRITE(*,*) '     * Electromagnetic Application Research LAB.                 *'
   WRITE(*,*) '     *  Microelectronics and Information Systems Research Center *'
   WRITE(*,*) '     *  If you have any questions, please to forward your        *'
   WRITE(*,*) '     *  message to rbhwang@eic.nctu.edu.tw                       *'
   WRITE(*,*) '     ============================================================'
   Write(*,*)

   Return
   End
```